

# EViews 使用入門

## 第一節 EViews 軟體基本介紹


EViews 是 Econometrics Views 的縮寫，是美國 Quantitative Micro Software 公司所生產的，以視窗圖形為介面，從 EViews1.0、EViews2.0、EViews3.1，到目前最新的 EViews4.0 上市，功能越來越強大，且使用介面也越來越簡便、人性化。其主要特色在進行時間數列分析及預測方面有很強的功能。本使用說明以 EViews 3.1 為例。

EViews 操作視窗可分為兩種：第一種具有視窗軟體視覺化操作的優良特性。可以使用滑鼠對標準的視窗功能表和對話方塊進行操作。操作結果會出現在視窗中並能對操作結果直接進行處理。此外，EViews 還擁有強大的命令功能和批次處理語言功能，除了可以在 EViews 的命令視窗內輸入、編輯和執行命令，還能在程式檔中建立和儲存程式碼，以便在後續的研究中使用這些程式。一般而言，使用文字模式進行研究通常較使用視窗模式有好處，當處理大量的資料時，程式模式能夠提供視窗模式所沒有辦法提供的功能，如迴圈。


## 第二節 如何啟動與執行 EViews 程式


### 如何啟動 EViews

EViews 啟動方法有 3 種，首先進入 Windows 畫面（如下圖）：





### 啟動方法一：

點選左下角的  鈕，再選擇 **程式集** → **Eviews 3** → **Eviews 3.1**。




## 啟動方法二：

可將啟動方法一的執行程式  拖曳至桌面形成捷徑，直接雙點桌面上的 EViews3.1 捷徑。




## 啟動方法三：

日後執行程式完，可將結果儲存成 EViews 工作檔案 (Workfile)，直接雙點桌面上 EViews 工作檔案 Example.wfl



## Eviews 介面說明

當完成啟動步驟後，便可見到以下畫面。接下來說明 EViews 的軟體介面。  
EViews 介面說明



EViews 視窗分為幾個部分：標題欄、主功能表、命令視窗、狀態行和工作區。

### 1、標題欄 ( The Title Bar )

標題欄位於主視窗的頂部，標記有 EViews 字樣。當 EViews 視窗處於啟動時，標題欄顏色加深，否則變暗。可以點擊 EViews 視窗的任意區域將使它處於啟動狀態。

### 2、主功能表 ( The Main Menu )

主功能表位於標題欄之下。將指標移至主功能表上的某個工具並用滑鼠左鍵點擊，打開一個下拉式功能表，通過點擊下拉功能表中的工具，就可以對它們執行功能。功能表中黑色的是可執行的，灰色的是不可執行的無效工具。

### 3、命令窗口 ( The Command Window )

主功能表下的區域稱作命令視窗。在命令窗口鍵入命令，按 ENTER 後命令立即執行。命令視窗中的一閃一閃位置則稱為插入點。它指示鍵盤鍵入字元的位置。

將插入點移至從前已經執行過的命令行，可以編輯已經存在的命令，按 ENTER，立即執行原命令的編輯版本。

命令視窗支援剪下和貼上功能，命令視窗 其他 EViews 視窗和其他 Windows 程式視窗間可以方便地進行內容的移動。命令視窗的內容可以直接保存到文字檔案中備用，為此必須保持命令視窗處於啟動狀態，並從主功能表上選擇 File/Save As。

若鍵入的命令超過了命令視窗顯示的大小，視窗中就自動出現捲軸，透過上下或左右調整，可瀏覽已執行命令的各個部分。將指標移至命令視窗下部，按著滑鼠左鍵向下或向上拖動，可以調整命令視窗的大小。

#### 4、狀態行 ( The Status Line )

視窗最底部是狀態行。狀態行分為 4 個欄。

左欄有時給出 EViews 送出的狀態資訊，點擊狀態行左端的邊框可以清除這些資訊。第二欄是 EViews 目前的讀取資料和程式的路徑。最後兩欄分別顯示目前資料庫和工作檔。

#### 5、工作區 ( The Work Area )

視窗的中心區域稱為工作區。EViews 在此顯示它建立的各種物件的視窗，本區也是使用時最常使用的區域，因為工作區中的這些視窗類似於使用者在辦公桌上用以工作的各類紙張。出現在最上面的視窗正處於焦點，即處於啟動狀態。狀態欄顏色加深的視窗是啟動視窗。


點擊部分處於下面的視窗的標題欄或任何可見部分，都可以使該視窗移至頂部。或者按壓 F6 或 CTRL-TAB，迴圈地啟動各個視窗。各視窗可以移動、改變大小。

### 第三節 EViews 資料的建立與讀取

在第一節已經說過 EViews 的使用可以分為兩種模式，分別為視窗模式與文字模式，其中文字模式又可分為直接在命令視窗鍵入文字與程式模式。本使用說明將集中在介紹如何使用程式模式，以便讓讀者也能夠利用 Eviews 進行研究分析。


#### 我的第一隻 EViews 程式

要開始以程式模式執行 EViews，請先開啟 EViews，待開啟完成後，選擇主功能表中 **File**→**New**→**Program**，開啟編寫程式視窗。



你可以使用自己喜歡的文字編輯程式撰寫程式，只要將副檔名存成 EViews 程式檔 (\*.prg) 即可，從 File→Open→Program，即可讀取程式進行修改。

接下來，進入程式視窗，開始撰寫第一隻程式如下：



**create first u 1 218**

這是寫任何一個 EViews 程式一開始都需要做的宣告，**create** 代表建立一個新程式工作檔，一定要這樣寫；**first** 代表為程式的工作檔(workfile)命名，使用者可自取；**u** 代表資料格式，分別為年資料(**a**)、半年(**s**)、季(**q**)、月(**m**)、週(**w**)、日(**d**)及不指定(**u**)，其中日資料又可分為一星期有五日與一星期有七日；**1 218** 代表觀察值有 218 個，一般只要設定比實際觀察值大即可。

以下以資料來源為 2002 年 1 月 1 日至 2004 年 12 月 31 日為例，分別簡要說明以各種資料格式要如何讀入。

```
create annual a 2002 2004/  
create semiannual s 2002:01 2004:04  
create quarter q 2002:01 2004:04  
create month m 2002:01 2004:12  
create week w 1/1/2002 12/31/2004  
create daily_5 d 1/1/2002 12/31/2004  
create daily_7 7 1/1/2002 12/31/2004
```

### EViews 小技巧


一般在撰寫程式的時後都不指定日期格式，拿日資料來說，一但指定，EViews 便會給予日期指定值，通常因為實際的交易日期會與電腦的指定值有所不同，所以反而會有所混淆。而星期資料則會以資料起始星期為資料星期，若讀入第一筆資料為星期三，則之後每個觀察值都會是星期三。

```
read(a2,s=sheet1) c:\eviews\ex1.xls date mk tsmc umc
```


這一行代表就是讀取外部 Excel 資料檔，在後面會有更詳細說明，c:\eviews\ex1.xls 代表讀取檔案的位置，date mk tsmc umc 代表讀入 4 個數列 (series) 變數，分別命名為 date、mk、tsmc 與 umc，也就是日期、台股加權指數、台積電收盤價與聯電收盤價。

等到程式撰寫完畢，請按程式左上方的 **Run** 執行程式，執行結果如下：






Workfile 即為儲存結果的區域，而 c 與 resid 為系統預設的係數，每一個 Workfile 都有，用來存放迴歸分析的常數項 (c) 與殘差項 (resid)，其餘的四個變數就是我們所要的資料，每一個變數皆可雙擊開啟資料內容。



如此一來就完成了第一隻程式，要記得把程式存檔喔！存檔的方法也非常簡單，只要按一下程式視窗上的 **Save** 就可以將程式命名存下。

若是想要一次檢視我們所讀入的所有資料，可以壓著鍵盤上的 **Ctrl** 鍵以滑鼠左鍵點選所要開啟的數列，點選完畢後按滑鼠右鍵，選擇 **Open**→**As Group** 開啟，即可同時檢視多組數列。



檢視結果如下：

Workfile: FIRST - (c:\program files\evIEWS\Workfile)

View | Proc | Objects | Save | Label+ | Show | Fetch | Store | Delete | Gear | Sample

Range: 1 218 Filter: \* Default Eq: None  
Sample: 1 218

Group: UNTITLED Workfile: FIRST

obs	DATE	MK	TSMC	UMC
1	20020103	5526.320	86.00000	49.90000
2	20020104	5638.530	85.50000	49.10000
3	20020107	5834.690	89.00000	52.00000
4	20020108	5810.080	92.50000	52.50000
5	20020109	5865.540	90.50000	50.50000
6	20020110	5871.280	91.50000	50.50000
7	20020111	5687.590	90.50000	48.90000
8	20020114	5611.860	87.50000	46.60000
9	20020115	5592.740	88.00000	45.60000
10	20020116	5488.330	87.50000	44.30000
11	20020117	5501.130	85.00000	43.00000
12	20020118	5522.800	84.50000	44.70000
13	20020121	5798.050	85.00000	45.30000
14	20020122	5804.100	90.50000	48.30000
15	20020123	5769.780	90.00000	47.50000
16				

### EViews 小技巧

其實在 Workfile 視窗下也可以儲存 Workfile 的結果，只是當我們在進行資料處理運算過程當中，會產生許多不必要的暫存資料，若是將 Workfile 的資料全部儲存，不免會浪費許多的空間。所以比較的好方法是盡量只儲存程式碼與所想要的資料即可。

### 資料建立步驟

#### 1、以純文字檔建立的文字資料檔

由於 EViews 並不支援 Word 的格式 (\*.doc)，故建立資料時請存成純文字檔，例如\*.txt、\*.csv 或\*.dat 等皆可。

#### 2、以 Excel 建立的文字檔案

由於一般進行資料分析與處理時可能會先將資料在 Excel 內進行簡單的處理，故 EViews 也可以讀取 Excel 檔案。

### 讀取外部資料與儲存

在 EViews 底下讀入檔案的方式有兩種，一種是以數列(series)讀入，而另一種是以矩陣(matrix)的型態讀入。兩者最大不同在於前者需要一欄一欄指定變數

名稱，而後者只須宣告一個矩陣變數即可。兩種不同方法請參考下面的範例 ( Sample1.prg )。

```
'--- Define the data name, data type and range ---'  
create first u 1 218  
  
'--- Input data as series ---'  
read(a2,s=sheet1) c:\evIEWS\ex1.xls date mk tsmc umc  
  
series mkre=dlog(mk)'--- calculate return ---'  
series tsmcre=dlog(tsmc)  
series umcre=dlog(umc)  
  
'===== '  
'=          another way          ='  
'===== '  
  
'--- Input data as matrix ---'  
matrix(218,4) mall  
mall.read(a2,s=sheet1) c:\evIEWS\ex1.xls  
  
'--- Extract the vector from the matrix ---'  
vector vmk=@columnextract(mall,2)  
vector vtsmc=@columnextract(mall,3)  
vector vumc=@columnextract(mall,4)  
  
'---Convert the vector to series ---'  
mtos(vmk,smk)  
mtos(vtsmc,stsmc)
```

```

mtos(vumc,sumc)
'--- Calculate return ---'
series smkre=dlog(smkc)
series stsmcre=dlog(stsmc)
series sumcre=dlog(sumc)

'--- Save the return data ---'
matrix(218,4) allre
matplace(allre,date,1,1)
matplace(allre,mkre,1,2)
matplace(allre,tsmcre,1,3)
matplace(allre,umcre,1,4)

allre.write c:\eviews\allre.txt

'--- Operation as loop ---'
matrix(218,4) allre2
for !a=1 to 4
  if !a=1 then
    matplace(allre2,date,1,1)
  else
    vector p!a=@columnextract(mall,!a)
    mtos(p!a,sp!a)
    series re!a=dlog(sp!a)
    matplace(allre2,re!a,1,!a)
  endif
next

```

在每一種程式當中，都會提供在程式碼中加入註解的功能，不但能夠增加程式的可讀性，也能夠紀錄一些事項。在 EViews 中註解的語法就是：'。只要加上'符號，之後的文字一律不會被讀入。分行說明如下。

```
read(a2,s=sheet1) c:\evIEWS\ex1.xls date mk tsmc umc
```

`read` 是一種讀入資料的語法，而因為資料為 Excel 檔，可以指定從哪一個開始欄位讀取，`a2` 就代表欄位，而 `sheet1` 代表讀入的工作表；`c:\evIEWS\ex1.xls` 代表存放檔案的路徑；`date mk tsmc umc` 代表讀入四數列，可自行命名。

	A	B	C	D	E	F	G
1	date	mkclos	tsmcclos	umcclos			
2	20020103	5526.32	86	49.9			
3	20020104	5638.53	85.5	49.1			
4	20020107	5834.89	89	52			
5	20020108	5810.08	92.5	52.5			
6	20020109	5865.54	90.5	50.5			
7	20020110	5871.28	91.5	50.5			
8	20020111	5687.59	90.5	48.9			
9	20020114	5611.86	87.5	46.6			
10	20020115	5592.74	88	45.6			
11	20020116	5488.33	87.5	44.3			
12	20020117	5501.13	85	43			
13	20020118	5522.8	84.5	44.7			
14	20020121	5798.05	85	45.3			
15	20020122	5804.1	90.5	48.3			
16	20020123	5769.78	90	47.5			

### EViews 小技巧

當要開啟 Excel 檔時，與 SAS 不同，必須要關閉 Excel，否則會無法讀取。或者可以將要讀入的 Excel 檔案屬性設為「唯讀」，便可以在讀取時開啟。

上面介紹讀取外部檔案為 Excel 時的方法，若我們的資料來源為文字檔，如.txt 時，讀取資料的方法為何？其實跟上面的方法十分雷同，只是不需要指定讀入 sheet 為何，程式如下：

```
read c:\evIEWS\ex1.txt date mk tsmc umc
```

接下來繼續說明有關資料處理的步驟。

```
series mkre=dlog(mk)
series tsmcre=dlog(tsmc)
```

上式為計算股價報酬，在 EViews 中，數列可以直接進行一些簡單運算，等會兒以矩陣讀入資料感受會更深，`dlog()`代表對數列的自然對數值取一階差分，此乃連續型報酬的計算方式，計算結果必須要存成一個新數列，`series mkpre`，就是宣告一個新的數列，數列名稱也可自取。另外，若是需要計算間斷型報酬，也可以利用 `series mkre=(mk-mk(-1))/mk(-1)`或 `series mkre=mk/mk(-1)-1`。

再來跟大家介紹以矩陣讀入的方式，請繼續往下看。

```
matrix(218,4) mall
```

先宣告一個矩陣，`matrix(218,4)`代表矩陣的大小，`rows` 為 218，`columns` 為 4；`mall` 為名稱。

```
mall.read(a2,s=sheet1) c:\leviews\lex1.xls
```

這以矩陣讀資料的方式，`mall.read` 前面代表矩陣名稱，`.read` 是方法，其餘同前面。因為要讀入日期資料，所以從 `a2` 開始讀入。同樣地，我們也能夠將文字檔以矩陣方式讀入，程式如下：

```
mall.read(skiprow=1) c:\leviews\lex1.txt
```

在這裡我們可以看到當我們對文字檔以矩陣方式讀入時，除了不需要指定 `sheet` 外，又多加了一個 `skiprow=1` 的選項敘述。由於我們的資料第一個 `row` 都會存放資料的名稱，但是當 Eviews 讀入資料時我們並不需要讀入文字的部分，所以我們便加上一個 `skiprow=1` 的選項以便從第二個 `row` 開始讀入。同理，當我們不需要讀入第一個 `column` 的日期資料時，也可以加上一個 `skipcol=1` 的選項即可，兩個選項之間以逗點隔開即可。

```
vector vmk=@columnextract(mall,2)
vector vtsmc=@columnextract(mall,3)
vector vumc=@columnextract(mall,4)
```

因為方程式必須以數列運算，所以必須把矩陣資料先粹取出，再進行轉換，`@columnextract(mall,2)`是一種從矩陣粹取一欄出來的函數，因為粹取出的矩陣只剩一行，所以也是一種向量，即 `vector`。

```
mtos(vmk,smk)
mtos(vtsmc,stsmc)
mtos(vumc,sumc)
```

而 `mtos` 就是一種將矩陣轉換為數列的函數，前面放矩陣名稱，後面放新產生數列名稱。

#### EViews 小技巧

根據使用經驗，Eviews 對於中文的辨識能力相當弱，所以不論是即將要被讀入的資料，或者是存放程式碼的路徑，最好不要在任何中文資料夾底下，有時會讀不出來，檔名也請不要以中文命名，資料內也請不要有中文欄位，一定會辨識不出來。不知道新版本有沒有改進以上缺點。

後面計算報酬的方法與前面相同，只是以不同名稱命名。接下來介紹如何將整理過的資料存出硬碟。

```
matrix(218,4)allre
```

同樣地，因為整理過後的資料可能需要保留下來，所以需告一個新的矩陣，名稱為 `allre` 來放置。

```
matplace(allre,date,1,1)
matplace(allre,mkre,1,2)
matplace(allre,tsmcre,1,3)
matplace(allre,umcre,1,4)
```

在這裡 `matplace` 的功用就是要將我們剛剛所計算出的報酬數列置入矩陣中，而 `matplace(allre,date,1,1)` 意義為將 `data` 置入，`allre` 中第一個 row，第一個 column 開始放起(`allre (1,1)`)。依序將所計算出的報酬數列全部置入。

```
allre.write c:\eviews\allre.txt
```

當我們把要存出的矩陣整理好後，就如同我們要從矩陣讀資料一樣利用 `.read` 方法，此時只要使用 `.write` 即可，`c:\eviews\` 代表存檔路徑，`allre.txt` 則代表所要存檔的名稱，副檔名不限為 `.txt`，但是都會以純文字檔的形式儲存。

以上這種資料處理方式，當你要一次處理的資料家數可能有上百公司甚至上



千家公司時，可能需要打字打到手軟了，所以接下來將介紹以迴圈<sup>1</sup>的方式來處理，同時也有一個判斷式的應用。

```
matrix(218,4)allre2
```

本部份將與前一部分作比較，所以同樣宣告一個名稱為 **allre2** 的矩陣。

```
for !a=1 to 4
  if !a=1 then
    matplace(allre2,date,1,1)
  else
    vector p!a=@columnextract(mall,!a)
    mtos(p!a,sp!a)
    series re!a=dlog(sp!a)
    matplace(allre2,re!a,1,!a)
  endif
next
```

在 EViews 中，「驚歎號+變數名稱」代表乃是一個暫存的變數，只會存在於運算過程，並不會在 Workfile 中留下，適合拿來當作迴圈的計數器；所以上例中的 **!a** 就是一個暫存計數器變數。而由於我們只需要重複從第二個 column 做到第四個 column，所以增加條件判斷式，當 **!a** 為 1 時，就直接把日期放在 **allre2** 內即可。待執行完畢之後可以比較看看 **allre** 和 **allre2** 的值是不是一樣。

## 第四節 EViews 中基本統計與資料處理

本部將介紹許多基本統計分析與資料處理的一些方法。

### 基本敘述統計量

進行統計分析第一步，就是要知道分析數值的基本係數統計量，EViews 也提供方法檢視，以下面的程式說明。（Sample2.prg）

---

<sup>1</sup> 有關迴圈的基本概念，請自行參考任何一本程式設計的書籍皆有說明，判斷式雷同，本部分只介紹在 EViews 上語法及應用。

```
'--- Define the data name, data type and range ---'  
create statistics u 1 218  
  
'--- Input data as series ---'  
read(a2,s=sheet1) c:\reviews\ex1.xls date mk tsmc umc  
  
'--- Calculate return ---'  
series mkre=dlog(mk)  
series tsmcre=dlog(tsmc)  
series umcre=dlog(umc)  
  
'--- Calculate descriptive statistics as scalar ---'  
scalar mkMin=@min(mkre)  
scalar mkQ25=@quantile(mkre,0.25)  
scalar mkMedian=@median(mkre)  
scalar mkQ75=@quantile(mkre,0.75)  
scalar mkMax=@max(mkre)  
  
scalar mkMean=@mean(mkre)  
scalar mkStdev=@stdev(mkre)  
scalar mkVar=@var(mk)
```

```
'--- Calculate descriptive statistics in table ---'
table(9,2) mkstats
setcell(mkstats,1,1,"mkMin")
setcell(mkstats,2,1,"mkQ25")
setcell(mkstats,3,1,"mkMedian")
setcell(mkstats,4,1,"mkQ75")
setcell(mkstats,5,1,"mkMax")

setcell(mkstats,7,1,"mkMean")
setcell(mkstats,8,1,"mkStdev")
setcell(mkstats,9,1,"mkVariance")

mkstats(1,2)=@min(mkre)
mkstats(2,2)=@quantile(mkre,0.25)
mkstats(3,2)=@median(mkre)
mkstats(4,2)=@quantile(mkre,0.75)
mkstats(5,2)=@max(mkre)

mkstats(7,2)=@mean(mkre)
mkstats(8,2)=@stdev(mkre)
mkstats(9,2)=@var(mkre)

'--- Using group calculate descriptive statistics ---'
group mygroup mkre tsmcre umcre
freeze(fmygroup) mygroup.stats
```

前面已經說明過的部分不再贅述，直接從沒說過的部分開始說明。

```
scalar mkMin=@min(mkre)
scalar mkQ25=@quantile(mkre,0.25)
scalar mkMedian=@median(mkre)
scalar mkQ75=@quantile(mkre,0.75)
scalar mkMax=@max(mkre)


scalar mkMean=@mean(mkre)
```

```
scalar mkStdev=@stdev(mkre)
scalar mkVar=@var(mkre)
```

在這邊，又有一個新的物件出現，叫做 **scalar**。望文生義，**scalar** 乃是指一個純量 一個單純的值而已；相對於 **series** 與 **vector** 一次可以包含多筆資料，**scalar** 只有一筆資料；相對於 **!a** 暫存於運算過程，**scalar** 可以被儲存在 Workfile 中；**scalar** 後面接變數的名字。


與我們之前所遇到的 **dlog(mk)** 相同，**@min(mkre)** 乃是代表一個函數，功能為求取 **mkre** 的最小值。同樣地，接下來的 **@quantile(mkre,0.25)** 是求 **mkre** 的第一分位差；**@median(mkre)** 是求 **mkre** 的中位數；**@quantile(mkre,0.75)** 是求 **mkre** 的第三分位差；**@max(mkre)** 是求 **mkre** 的最大值；**@mean(mkre)** 是求 **mkre** 的平均數；**@stdev(mkre)** 是求 **mkre** 的標準差；**@var(mkre)** 是求 **mkre** 的變異數。


以上這些函數指數 EViews 對於敘述統計量的一些函數，還許有多的運算函數，使用者可自行由 help 中查詢，只要按下主功能表中的 **Help**→**Function Reference**，就有非常多的函數，包括加減乘除、指數對數、四捨五入、與各種機率分配函數如何使用皆有說明。






我們可以先看看程式的執行結果。



我們可以發現到 Workfile 中多了許多  的變數，就是 scalar 的意思，現在可以雙擊你所想知道的敘述統計量，那麼就會在下方的狀態欄顯示出來，



進行到這裡可能會想說，那我要敘述統計量不就要自己將它抄寫出來嗎？這樣會不會太麻煩了，當然EViews是不需要這麼麻煩啦！底下就是要來教大家EViews中建立好表格，再將我們所需要的敘述統計量置入表格內。

```

table(9,2) mkstats
setcell(mkstats,1,1,"mkMin")
setcell(mkstats,2,1,"mkQ25")
setcell(mkstats,3,1,"mkMedian")
setcell(mkstats,4,1,"mkQ75")
setcell(mkstats,5,1,"mkMax")

setcell(mkstats,7,1,"mkMean")
setcell(mkstats,8,1,"mkStdev")
setcell(mkstats,9,1,"mkVariance")
    
```


由於我們需要在Workfile中建立一個表格來放置所需要的結果，所以便需要使用表格物件，表格的使用方法則如上table(9,2) mkstats；一樣地，「物件的種類+變數的名稱」，是宣告一個變數的基本原則，但是因為表格需要指定大小，所以table(9,2) mkstats乃是代表這個叫做mkstats的表格有9個row與2個column。

為什麼要設定2個column，是因為我們需要在第一個column擺放說明文字，這時候，setcell()這個函數就派上用場了，setcell(mkstats,1,1,"mkMin")代表要設定mkstats這個表格內的第(1,1)個欄位(cell)，前面是row，後面是column，由於此時需要放置的是字串(string)，所以需要使用代表文字的符號""，"mkMin"則表示要放在欄位內的文字為mkMin。以下的欄位以此類推。

```
mkstats(1,2)=@min(mkre)
mkstats(2,2)=@quantile(mkre,0.25)
mkstats(3,2)=@median(mkre)
mkstats(4,2)=@quantile(mkre,0.75)
mkstats(5,2)=@max(mkre)

mkstats(7,2)=@mean(mkre)
mkstats(8,2)=@stdev(mkre)
mkstats(9,2)=@var(mkre)
```

當我們將文字欄位設定完畢後，接下來就要將所計算的敘述統計量也放入表格中，比較簡單的方法就是直接使用 `mkstats(1,2)=@min(mkre)`，意思就是令mkstats的第(1,2)個欄位等於mkre的最小值(@min(mkre))。其餘的統計量類推。聰明的你是不是想問問可不可以用上面的方法，改用setcell(mkstats,1,2,@min(mkre))，當然也是可以啦！不過這樣好像要打的字數比較多。待所有統計量設置完畢後，就可以執行程式，執行完畢後直接打開Workfile的 `mkstats`，便可以欣賞到剛剛自己設定的結果了。



以下，將提供另一個可以讓使用者一次檢視許多數列基本敘述統計量的方法。

### group mygroup mkre tsmcre umcre

在這裡，可以利用一個新的物件 **group** 將所需要的數列成為一組群組，以方便同時進行分析，**group mygroup mkre tsmcre umcre** 代表宣告一個 **group** 物件，名稱為 **mygroup**，後面接的就是群組的成員。


### freeze(fmygroup) mygroup.stats

EViews 本身就有提供檢視敘述統計量的方法，就是在數列或群組的名稱後面加上 **.stats**。這個方法最大的問題在於結果只能供你檢視暫時檢視，並不會儲存在 Workfile 中。這時候就要使用一點小技巧，利用 **freeze** 這個函數，**freeze** 的功能在於可以把使用者想要保留下的狀態給凍結起來，產生表格或圖形以供使用者作進一步利用。在 EViews 中許多的統計檢定在程式的執行過程當中並不會將結果儲存起來，若是一連利用程式作了許多檢定，便只能夠檢視到最後一個統計檢定的結果而已，這時候 **freeze** 這個函數就可以它的效果，若是能夠在每一個檢定之後再利用 **freeze** 指令，便不會遺漏掉每一個統計檢定的結果了。

**freeze(fmygroup) mygroup.stats** 代表產生一個名稱為 **fmygroup** 的表格，來源是將 **mygroup.stats** 給凍結，而 **.stats** 則是檢視數列基本敘述統計量的方法。程



式執行完畢後直接檢視 Workfile 中的 Table fmygroup，便可以同時知道 group 成員內的敘述統計量了，至於有些敘述統計量可能裡面沒有，如果有需要，還是得靠使用者自行增加了。



## 第五節 EViews 迴歸分析

在這部分將介紹如何利用 EViews 進行迴歸分析。

### 相關係數

在進行迴歸分析之前需先檢查自變數之間的相關係數，是具有高度相關性，則迴歸分析會有共線性的問題，必須先剔除某些變數才能繼續進行分析。本部分以「財金計量」第三章範例<sup>2</sup>說明，以台灣、美國與日本股市為例。說明如下：

<sup>2</sup> 「財金計量」第 81 頁，光碟檔案 c:\floppy\ch03\data\reg.dat 資料加以修改為 reg.txt。

( Sample3.prg )

```
'--- Read data from txt file ---'  
create reg u 1 39  
read c:\evIEWS\reg.txt date tw us jp  
  
'--- View correlation coefficient ---'  
group mygroup tw us(-1) jp(-1)  
mygroup.cor
```

接下來，我們就來看看如何計算相關係數。

```
group mygroup tw us(-1) jp(-1)  
mygroup.cor
```

由於我們需要計算相關係數矩陣，所以將三個國家的股價報酬設為一個群組，由於時差與交易時間的關係，我們認為美、日股市對台股會延後一日產生影響，故形成群組時需將美、日股市延後一日，在這裡 **us(-1)**代表 **us** 數列遞延一期，當然也可以利用 **series uslag=us(-1)**產生一個新數列。

如同**.stats**代表求敘述統計量的方法一樣，**mygroup.cor**代表一個新的方法，就是求相關係數矩陣。待執行程式時，就會自動跳出結果如下。

Group: MYGROUP Workfile: REG

View Procs Objects Print Name Freeze Sample Sheet Stats Spec

Correlation Matrix

	TW	US(-1)	JP(-1)	
TW	1.000000	0.344190	0.055058	
US(-1)	0.344190	1.000000	0.009137	
JP(-1)	0.055058	0.009137	1.000000	

這樣就可以看出來台股與美、日股的相關性，在本例中並沒有共線性的問題。

### 簡單迴歸分析

在確認過沒有共線性後，便可以進行迴歸分析了，程式如下。




```
'--- Read data from txt file ---'
create reg u 1 39
read c:\evIEWS\reg.txt date tw us jp


'--- Regression analysis---'
equation eq01.ls tw c us(-1) jp(-1)
equation eq02.ls tw us(-1) jp(-1)
```

程式說明如下：

```
equation eq01.ls tw c us(-1) jp(-1)
equation eq02.ls tw us(-1) jp(-1)
```

在這裡，又看到一個新的物件 **equation**，顧名思義就是代表方程式的物件，後面接變數的名字；至於 **eq01.ls** 代表方程式 eq01 所使用的估計方式為 least

quare (ls) ; 等到名字與估計方法宣告完畢後，後面接應變數，如果需要截距項，請加上 c，如果不需要，就直接加上自變數。如此一來便完成迴歸式的估計，等到程式執行完畢後，開啟 Workfile 中  eq01 和  eq02 估計結果，若是使用者希望程式執行完畢時能夠自動開啟迴歸分析的結果，可以加上 **show eq01** 與 **show eq02** 兩個指令，那麼程式就會自動開啟 eq01 與 eq02 兩個方程式。開啟  eq01 結果如下。




Variable	Coefficient	Std. Error	t-Statistic	Prob.
C	-8.28E-05	0.002408	-0.034396	0.9728
US(-1)	0.457842	0.211084	2.169004	0.0370
JP(-1)	0.049982	0.152560	0.327624	0.7451


  

R-squared	0.121162	Mean dependent var	-0.000375
Adjusted R-squared	0.070942	S.D. dependent var	0.015300
S.E. of regression	0.014747	Akaike info criterion	-5.519891
Sum squared resid	0.007612	Schwarz criterion	-5.390608
Log likelihood	107.8779	F-statistic	2.412652
Durbin-Watson stat	1.806731	Prob(F-statistic)	0.104329

有時候當我們在進行迴歸分析時可能會需要將估計式的某些參數拿出來使用，除了一個一個抄出之外，EViews 有更好的方法。以上式為例，若是想要知道 us 的估計參數，可以使用 `scalar uscoef=eq01.@coefs(2)`，便可以產生一個名為 uscoef 的 scalar，內容為 us 的迴歸估計係數。.@coefs(2) 代表存取方程式物件的方法，前面加上方程式的名字，.@coefs() 乃是代表估計參數的意思，因為美國乃是所估計的第二個係數（常數項為第一個），故填入 2。再以調整後判定係數（adjusted R-squared）為例，方法為.@rbar2。更多更詳細的說明與參數請使用 help，從主功能表中的 **Help**→**Object Reference**，選擇 Equation



進入後將選單下拉至 **Equation Data Members**，即可查詢更多參數如何使用。



## 進階迴歸分析

接下來舉兩個比較特殊的迴歸模型，分別是 censored regression model 與 truncated regression model。簡言之，此兩種模型都是屬於應變數受限制模型 (limited-dependent variable model)，如股價受到漲跌停板的限制，所以我們所見到的股價其實都是在一定範圍內，超過 (小於) 停板的價格都見不到，只能見到停板的價格，就是屬於一種 censored sample；若是一旦觸及臨界值，便連臨界值也觀察不到，便是屬於 truncated sample。以下僅說明兩種模型的估計方式，至於應於請使用者自行練習。

**Censored : equation 方程式名稱.censored(l=left censored limit,r=right censored limit) 應變數 截距項(可整略) 自變數1 自變數2 ...**

與最小平方法不同，估計方法時要使用.censored，後面還要加上選項。其中，left censored limit 代表如跌停板股價，而 right censored limit 代表如漲停板股價

**Truncated : equation 方程式名稱 .censored(t,l=left censored limit,r=right censored limit) 應變數 截距項(可整略) 自變數1自變數2 ...**

Truncated 與 censored 估計方法唯一不同處在於 truncated 需要在選項處加上 t；除此之外，其餘與 censored 同。